#.rst:
# FindPulseAudio
# --------------
# Finds the PulseAudio library
#
# This will define the following variables::
#
#  PULSEAUDIO_FOUND - system has the PulseAudio library
#  PULSEAUDIO_INCLUDE_DIRS - the PulseAudio include directory
#  PULSEAUDIO_LIBRARIES - the libraries needed to use PulseAudio
#  PULSEAUDIO_DEFINITIONS - the definitions needed to use PulseAudio
#
# and the following imported targets::
#
#   PulseAudio::PulseAudio   - The PulseAudio library

if(PKG_CONFIG_FOUND)
  pkg_check_modules(PC_PULSEAUDIO libpulse>=11.0.0 QUIET)
  pkg_check_modules(PC_PULSEAUDIO_MAINLOOP libpulse-mainloop-glib>=11.0.0 QUIET)
  pkg_check_modules(PC_PULSEAUDIO_SIMPLE libpulse-simple>=11.0.0 QUIET)
endif()

find_path(PULSEAUDIO_INCLUDE_DIR NAMES pulse/pulseaudio.h pulse/simple.h
                                 PATHS ${PC_PULSEAUDIO_INCLUDEDIR} ${PC_PULSEAUDIO_INCLUDE_DIRS})

find_library(PULSEAUDIO_LIBRARY NAMES pulse libpulse
                                PATHS ${PC_PULSEAUDIO_LIBDIR} ${PC_PULSEAUDIO_LIBRARY_DIRS})

find_library(PULSEAUDIO_SIMPLE_LIBRARY NAMES pulse-simple libpulse-simple
                                PATHS ${PC_PULSEAUDIO_LIBDIR} ${PC_PULSEAUDIO_LIBRARY_DIRS})

find_library(PULSEAUDIO_MAINLOOP_LIBRARY NAMES pulse-mainloop pulse-mainloop-glib libpulse-mainloop-glib
                                         PATHS ${PC_PULSEAUDIO_LIBDIR} ${PC_PULSEAUDIO_LIBRARY_DIRS})

if(PC_PULSEAUDIO_VERSION)
  set(PULSEAUDIO_VERSION_STRING ${PC_PULSEAUDIO_VERSION})
elseif(PULSEAUDIO_INCLUDE_DIR AND EXISTS "${PULSEAUDIO_INCLUDE_DIR}/pulse/version.h")
  file(STRINGS "${PULSEAUDIO_INCLUDE_DIR}/pulse/version.h" pulseaudio_version_str REGEX "^#define[\t ]+pa_get_headers_version\\(\\)[\t ]+\\(\".*\"\\).*")
  string(REGEX REPLACE "^#define[\t ]+pa_get_headers_version\\(\\)[\t ]+\\(\"([^\"]+)\"\\).*" "\\1" PULSEAUDIO_VERSION_STRING "${pulseaudio_version_str}")
  unset(pulseaudio_version_str)
endif()

include(FindPackageHandleStandardArgs)
find_package_handle_standard_args(PulseAudio
                                  REQUIRED_VARS PULSEAUDIO_LIBRARY PULSEAUDIO_MAINLOOP_LIBRARY PULSEAUDIO_SIMPLE_LIBRARY PULSEAUDIO_INCLUDE_DIR
                                  VERSION_VAR PULSEAUDIO_VERSION_STRING)

if(PULSEAUDIO_FOUND)
  set(PULSEAUDIO_INCLUDE_DIRS ${PULSEAUDIO_INCLUDE_DIR})
  set(PULSEAUDIO_LIBRARIES ${PULSEAUDIO_LIBRARY} ${PULSEAUDIO_MAINLOOP_LIBRARY} ${PULSEAUDIO_SIMPLE_LIBRARY})
  set(PULSEAUDIO_DEFINITIONS -DHAS_PULSEAUDIO=1)

  if(NOT TARGET PulseAudio::PulseAudio)
    add_library(PulseAudio::PulseAudio UNKNOWN IMPORTED)
    set_target_properties(PulseAudio::PulseAudio PROPERTIES
                                                 IMPORTED_LOCATION "${PULSEAUDIO_LIBRARY}")
	target_include_directories(PulseAudio::PulseAudio INTERFACE ${PULSEAUDIO_INCLUDE_DIR})
	target_compile_definitions(PulseAudio::PulseAudio INTERFACE HAVE_LIBPULSE=1 _REENTRANT=1)
  endif()
  if(NOT TARGET PulseAudio::PulseAudioMainloop)
    add_library(PulseAudio::PulseAudioMainloop UNKNOWN IMPORTED)
    set_target_properties(PulseAudio::PulseAudioMainloop PROPERTIES
                                                         IMPORTED_LOCATION "${PULSEAUDIO_MAINLOOP_LIBRARY}")
	target_link_libraries(PulseAudio::PulseAudioMainloop INTERFACE PulseAudio::PulseAudio)
  endif()
  if(NOT TARGET PulseAudio::PulseAudioSimple)
    add_library(PulseAudio::PulseAudioSimple UNKNOWN IMPORTED)
    set_target_properties(PulseAudio::PulseAudioSimple PROPERTIES
                                                       IMPORTED_LOCATION "${PULSEAUDIO_SIMPLE_LIBRARY}")
	target_link_libraries(PulseAudio::PulseAudioSimple INTERFACE PulseAudio::PulseAudio)
  endif()
endif()

mark_as_advanced(PULSEAUDIO_INCLUDE_DIR PULSEAUDIO_LIBRARY PULSEAUDIO_MAINLOOP_LIBRARY PULSEAUDIO_SIMPLE_LIBRARY)
